KISSsoft AG - +41 55 254 20 50
Uetzikon 4 - +41 55 254 20 51
8634 Hombrechtikon - info@KISSsoft.AG
Switzerland - www.KISSsoft.AG

KISSsys Instruction:

Export of Bearing Forces

1 Introduction

A text file should be exported from a KISSsys model with all actual bearing forces. It is possible to use that file as input data for a FEM calculation of a gear box casing or for the check of the calculations (sum of the reaction forces = sum of the outer forces + weight of the components, if considered). The output forces are the forces from the shaft on the bearings (acting on the casing), not the reaction forces of the bearings.

2 Solution

The function „GetReactions“ collects every KISSsys elements of the type bearing („kSysBearing plus kSysRollerBearing“) and reads their bearing forces. These are gathered and written in a variable of the type text. This variable is then written in a file which is compatible e.g. to Excel. Finally those forces can be manipulated in Excel. Also other possible file formats are available.

3 Modelling

3.1 Using the Function

3.1.1 Inclusion into the Model

A new function can be introduces in the model e.g. under „System“, with a given name e.g. „GetReactions“.



Figure 3.1-1 Inclusion of the Function "GetReactions"

With a right click on “System/Properties” and „GetReactions/Edit” it is possible to edit the function:



Figure 3.1-2 Command for the Variable Edition

The function editor appears, where the text of the function can be introduced using „Copy/Paste“ (see in chapter 3.2 for a function code text). The function has to be compiled and saved and then the window closed with “Quit”.



Figure 3.1-3 Function Editor

That procedure defined the function in the KISSsys model.

3.1.2 Function Call in the “UserInterface”

A right click on the appropriate cell and the choice of „Insert function“ defines a function call.



Figure 3.1-4 Entering of a Function Call in a User Interface

A dialog appears and the user defines the name and the call of the function in it:



Figure 3.1-5 Function Call

The function is now called by a double click on the bright blue cell „Show Reaction Forces“ in the user interface.

3.1.3 Output of the Results

Before the bearing forces can be exported they have to be calculated. A kinematics calculation is not enough for the calculation of the bearing forces. The KISSsoft shaft calculations have to be made. Those calculations are called by a right click on „System” and the choice of „kSoftCalculate” After the shaft calculation created results, the function „Show Reaction Forces“ will first ask a name for the file, where results will be written and also the type of the file.



Figure 3.1-6 File name and type selection

After selections function will generate following message with every bearing forces and global coordinates.



Figure 3.1-7 Message at the Export of the Bearing Forces

This data is now written in a file named „Reactions.csv“, which can be found in the project folder. It is possible to open the file in Excel. At the import, excel creates the sum of all the bearing forces in the 3 dimensions.



	A	B	C	D	E	F	G	H
1	Forces acting from the bearings on the casing							
2								
3	Bearing Nam	Global x	Global y	Global z	Force Fx	Force Fy	Force Fz	
4								
5								
6	^.Rotor.Shaft	-199.5	36.5	0	-413.41	0	-1439.04	
7	^.Rotor.Shaft	-199.5	67.5	0	-470.43	0	-1637.53	
8	^.Rotor.Shaft	-161.88	36.5	-32.74	413.4	0	1439.04	
9	^.Rotor.Shaft	-161.88	67.5	-32.74	470.42	0	1637.53	
10	^.Rotor.Shaft	-199.5	217.62	23.15	-7660.46	0	-6031.81	
11	^.Rotor.Shaft	-199.5	260.37	23.15	5213.56	628.6	2497.51	
12	^.Rotor.Shaft	-199.5	143	97.15	2523.95	-1629.15	0	
13	^.Rotor.Shaft	-199.5	143	-59.85	4569.48	1000.51	4319.83	
14	^.Rotor.Shaft	-199.5	224	85.28	-106.88	0	-49.84	
15	^.Rotor.Shaft	-199.5	254	85.28	-58.31	0	-27.19	
16	^.Rotor.Shaft	-161.88	219.5	-32.74	-2931.98	0	-495.84	
17	^.Rotor.Shaft	-161.88	275.5	-32.74	-1623.78	0	-387.95	
18	^.Rotor.Shaft	-199.5	616.5	-32.74	-30.27	50.86	232.56	
19	^.Rotor.Shaft	-199.5	591.5	-32.74	-20.59	0	-34.94	
20	^.Rotor.Shaft	-224	642	-32.74	50.86	-62.55	-181.47	
21	^.Rotor.Shaft	-163	642	-32.74	0	11.69	-16.14	
22	^.Rotor.Shaft	-199.5	255	-32.74	41.89	0	89.82	
23	^.Rotor.Shaft	-199.5	273	-32.74	41.89	0	89.82	
24								
25	Sum of all reactions, Fxtot, Fytot, Fztot				9.34	-0.04	4.36	
26								
27								
28								
29								
30								

Figure 3.1-8 Output Table

3.2 Structure of the Function

3.2.1 Source Code:

```
VAR Supports,Support,results,Fxtot,Fytot,Fztot,forcevector,vectortot,coord,
Excelsummex,Excelsummey,Excelsummez,n,res, form, sep,File;

res = CADH_VarDialog(["Define ouput file name for reactions",300,100,0.3,1],
                     [C:VDLG_Str,"File name:", File],
                     [C:VDLG_StrCom,"File format:", [".csv",".txt",".rtf",".xls"],[0],0]);


IF res[0] THEN
File=res[1];
form =res[2];

sep = "";
IF form = ".csv" OR ".txt" THEN
        sep = ";";
ELSE
        sep = "   ";
ENDIF

Supports=_O.OBJ_GetChildren(1,"kSysBearing");

results=" ";
results=results+"Forces acting from the bearings on the casing"+"\n"+"\n";
results=results+"Bearing Name"+sep+"Global x-coord."+sep+"Global y-coord."+sep+"Global z-coord."+sep+"Force Fx
[N]"+sep+"Force Fy [N]"+sep+"Force Fz [N]\n\n";
vectortot={0,0,0};

FORALL Supports Support DO
        forcevector=Support.v_g({Support.Fx,Support.Fy,Support.Fz});
        //results=results+@KISSsys.GetPathTo(OBJ_GetPathTo(Support))+sep;
        results=results+Support.OBJ_GetName()+sep;
        coord=Support.l_g({0,0,0});
        results=results+CADH_ValToStr(CADH_Round(coord:x,2))+sep+CADH_ValToStr(CADH_Round(coord:y,2))
+sep+CADH_ValToStr(CADH_Round(coord:z,2))+sep;
        results=results+CADH_ValToStr(CADH_Round(forcevector:x,2))+sep+CADH_ValToStr(CADH_Round(forcev
ector:y,2))+sep+CADH_ValToStr(CADH_Round(forcevector:z,2))+"\n";
        vectortot=vectortot+forcevector;
NEXT

n=LEN(Supports);

Excelsummex="=summe(E5:E"+CADH_ValToStr(n+4)+")"+sep;
Excelsummey="=summe(F5:F"+CADH_ValToStr(n+4)+")"+sep;
Excelsummez="=summe(G5:G"+CADH_ValToStr(n+4)+")"+sep;

CADH_Message("The following data is written to the file " + File +form+":\n----------------------------------------------------
----------------------------\n\n"+results);

results=results+"\n";
results=results+"Sum of all reactions, Fxtot, Fytot, Fztot";
results=results+sep+sep+sep+sep+Excelsummex+Excelsummey+Excelsummez+"\n";

CADH_WriteToFile(kSoft_GetDir("*.ks")+"\\"+File+form,results);

ENDIF
```

3.2.2 Explanations

Command	Description
VAR	Declaration of the local variables (all types)
File	The name of the file
form	Selected file extension
sep	Type of separator between fields, depends on file extension selection
Supports=_O.OBJ_GetChildren(1,"kSysBearing")	Every element of the type “kSysBearing” is searched after and saved in the variable „Supports”
results=	The variable „results“ contains the output text. The output text is added line by line (results=results+...)
„\n“	Break
FORALL	Defines a loop for every element in the array „Supports“.
Support.v_g({Support.Fx,Support.Fy,Support.Fz})	The force components of the bearings are transferred from local to global coordinates.
@KISSsys.GetPathTo(OBJ_GetPathTo(Support))	Calls the name of the bearing.
coord=Support.l_g({0,0,0});	The coordinates of the bearing center are read.
CADH_ValToStr	Transfers a value to a string
CADH_Round	Rounds a value
N=LEN(Supports)	Defines the number of elements of the array support (the number of bearings) in the variable n.
Excelsummex="=summe(E6:E"+CADH_ValToStr(n+5)+");";	The introduction of the Excel formulas
CADH_WriteToFile(kSoft_GetDir("*.ks")+"\\"+File+form,results);	Exporting the file